Solutions for *Mathematica* Exercises (© P. Fischer)

Exercise 1: Numbers

In[1]:=
```
Sqrt[2] (* By default, Mathematica stays with exact expressions *)
```
Out[1]= $\sqrt{2}$

In[2]:=
```
N[%, 100] (* Force a numberical evaluation of the previous (%) line *)
```
Out[2]=
```
1.4142135623730950488016887242096980785696718753769480731766797379907324784621070388503875343276415727
```

In[3]:= $\frac{20!}{45}$

Out[3]= 54 064 489 070 592 000

In[4]:= $\frac{20!}{46} \in$ Integers (* We can check whether the result is an integer *)

Out[4]= False

In[5]:=
```
FactorInteger[20!] (* FactorInteger[...]
 generates a list of prime factors with multiplicities *)
```
Out[5]=
```
{{2, 18}, {3, 8}, {5, 4}, {7, 2}, {11, 1}, {13, 1}, {17, 1}, {19, 1}}
```

In[6]:=
```
Sqrt[-4]
```
Out[6]= 2 ⅈ

In[7]:= F1 = $\frac{4}{5}$; F2 = $\frac{5}{8}$;

In[8]:=
```
F1 + F2
```
Out[8]= $\frac{57}{40}$

In[9]:=
```
F1 * F2
```
Out[9]= $\frac{1}{2}$

In[10]:=
```
Sqrt[F1]
```
Out[10]= $\frac{2}{\sqrt{5}}$

In[11]:= $1 - \frac{\sqrt{2\pi 100}\left(\frac{100}{e}\right)^{100}}{100!}$ // N

Out[11]= 0.000832983

In[12]:= $1 - \frac{\sqrt{2\pi n}\left(\frac{n}{e}\right)^{n}}{n!}$ /. n → 100 // N (* more general,
replacing n by a fixed value, see later *)

Out[12]= 0.000832983

Exercise 2: Lists and Expressions

In[13]:= alist = Table[n^2, {n, 1, 10}]

Out[13]= {1, 4, 9, 16, 25, 36, 49, 64, 81, 100}

In[14]:=
```
{First[alist], alist[[3]], Last[alist]}
```
Out[14]= {1, 9, 100}

In[15]:=
```
v1 = {1, 2, 3}; v2 = {2, 3, 4};
```
In[16]:=
```
v1.v2
```
Out[16]= 20

In[17]:=
```
A = {{1, 2}, {3, 4}}
```
Out[17]= {{1, 2}, {3, 4}}

In[18]:=
```
A // MatrixForm
```
Out[18]//MatrixForm=
$$\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$$

In[19]:=
```
B = Inverse[A]
```
Out[19]= $\left\{\{-2, 1\}, \left\{\frac{3}{2}, -\frac{1}{2}\right\}\right\}$

In[20]:=
```
A.B // MatrixForm
```
Out[20]//MatrixForm=
$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

In[21]:= $A = 3x + 5;\ B = 5x^2 - 7;$

In[22]:=
```
F = A B
```
Out[22]= $(5 + 3x)(-7 + 5x^2)$

In[23]:=
```
F // Expand
```
Out[23]= $-35 - 21x + 25x^2 + 15x^3$

In[24]:=
```
F // Factor
```
Out[24]= $(5 + 3x)(-7 + 5x^2)$

In[25]:=
```
F / A
```
Out[25]= $-7 + 5x^2$

In[26]:=
```
F = F // Expand
```
Out[26]= $-35 - 21x + 25x^2 + 15x^3$

In[28]:=
```
F / A (* When F is expanded, the fraction cannot be simplified so easily *)
```
Out[28]=
$$\frac{-35 - 21x + 25x^2 + 15x^3}{5 + 3x}$$

```
% // Simplify (* Simplify finds the simpler expression *)
```

Out[29]= $-7+5\,x^2$

In[30]:=
```
13.3456 // FractionalPart
  (* FractionalPart gives the digits after the comma *)
```

Out[30]= 0.3456

```
N[Table[e^(π √k), {k, 160, 170}], 100] // FractionalPart
  (* For k=163, the expression is very clos to an integer *)
```

Out[33]=
```
{0.6252838975958655381398604415281768115150431571225498936099564726436700894\
   822585298,
 0.6916517385923929696993549222375465501012527316155787763269082001206791146\
   059590332,
 0.9433718537913400160690150242592267896097954528496994384086455059617570193\
   056970892,
 0.9999999999992500725971981856888793538563373369908627075374103782106479101\
   186073130,
 0.0026726248354647230571999609920977505604998845455156191082210558332366612\
   668738022,
 0.6717729300164256924504109797765040155253396060204453970742299676171458579\
   622545777,
 0.6323410215714765810314272693733100139026773166163361312854204703769793580\
   062714250,
 0.1112246629953976549854325016657502678318904561610453265785888077071226468\
   683465500,
 0.7126569549154415668735952387959408801335614655054155873899708514495587336\
   381178697,
 0.4624638128351570543371097717810871880389776992792956791023714634582890285\
   572593878,
 0.7785477606206950415043551707464719261160938757685293771674984829221085777\
   071170072}
```

Warmup

In[34]:= $\text{Sum}\left[\frac{1}{i^2}, \{i, 1, \infty\}\right]$

Out[34]= $\frac{\pi^2}{6}$

In[37]:=
```
Sum[1/i^2, {i, 1, 100}]
```
Out[37]=
```
1 589 508 694 133 037 873 112 297 928 517 553 859 702 383 498 543 709 859 889 432 834 803 ⋱
  818 131 090 369 901 /
 972 186 144 434 381 030 589 657 976 672 623 144 161 975 583 995 746 241 782 720 354 705 ⋱
  517 986 165 248 000
```

In[38]:=
```
Sqrt[6 × %] // N
```
Out[38]=
```
3.13208
```

In[39]:=
```
%/π
```
Out[39]=
```
0.996971
```

In[41]:=
```
Table[Sin[α], {α, 0, π, π/4}]
```
Out[41]=
$\left\{0, \frac{1}{\sqrt{2}}, 1, \frac{1}{\sqrt{2}}, 0\right\}$

In[42]:=
```
Table[i, {i, 1, 10}]
```
Out[42]=
```
{1, 2, 3, 4, 5, 6, 7, 8, 9, 10}
```

In[45]:=
```
L = Range[10] (* This is the same. Quite useful *)
```
Out[45]=
```
{1, 2, 3, 4, 5, 6, 7, 8, 9, 10}
```

In[44]:=
```
CalculateSquare[x_] = x^2;
```
```
L // CalculateSquare (* Postix *)
```
Out[46]=
```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

In[48]:=
```
CalculateSquare[L] (* Call function an 'thread' though list *)
```
Out[48]=
```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

In[49]:=
```
CalculateSquare @ L (* Prefix *)
```
Out[49]=
```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

In[51]:=
```
#^2 & @ L (* Using pure function '#^2&' *)
```
Out[51]=
```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

Exercise 3: Solving Equations

Am Tisch sitzen die 3 Freunde Andy, Bob, Conny
- Conny ist 2 Jahre älter als Andy
- Conny ist doppelt so alt wie Bob
- Zusammen sind sie 38 Jahre alt

```
Clear[A, B, C]; (* Expressions have been used before... *)
```
Clear: Symbol C is Protected.

```
Solve[A + B + C == 38 && 2 B == C && C == A + 2, {A, B, C}] //
 First (* unique solution *)
```
Out[55]=
```
{A → 14, B → 8, C → 16}
```

In[56]:=
```
Solve[A + B + C == 38 && 2 B == C && C > A, {A, B, C}] (* With 'C>A' only a range *)
```
Solve: Equations may not give solutions for all "solve" variables.

Out[56]=

$$\left\{\left\{A \to \text{ConditionalExpression}\left[38 - \frac{3\,C}{2},\ C > \frac{76}{5}\right],\ B \to \text{ConditionalExpression}\left[\frac{C}{2},\ C > \frac{76}{5}\right]\right\}\right\}$$

In[57]:=
```
Reduce[A + B + C == 38 && 2 B == C && C > A, {A, B}]
(* Reduce gives a 'nicer' result (but content is the same) *)
```
Out[57]=

$$C > \frac{76}{5}\ \&\&\ A == \frac{1}{2}(76 - 3\,C)\ \&\&\ B == 38 - A - C$$

In[58]:=
```
Reduce[A + B + C == 38 && 2 B == C && C > A && A > 0 && B > 0 && C > 0, {A, B}]
(* Asking for positive ages restricts C even more *)
```
Out[58]=

$$\frac{76}{5} < C < \frac{76}{3}\ \&\&\ A == \frac{1}{2}(76 - 3\,C)\ \&\&\ B == 38 - A - C$$

Exercise 4: Simple Plotting

In[60]:=
```
f1[k_, x_] := Sin[k x];
```

In[62]:=
```
Plot[f1[1, x], {x, 0, 2 π}, Frame → True, ImageSize → 300, Filling → Axis]
```

Out[62]=



In[63]:=
```
Plot[f1[k, x] /. k → {1, 2}, {x, 0, 2 π},
 Frame → True, ImageSize → 300, Filling → Axis]
```

Out[63]=



In[65]:=
```
Plot[{Sin[x], Cos[x]}, {x, 0, 4 π}, Frame → True, ImageSize → 300, Filling → Axis]
```

Out[65]=



Exercise 5: Solving Equations

In[66]:=
```
f1[x_] := 3 x + 2; f2[x_] = 2 (x - 2)^2;
```

In[67]:=
```
Plot[{f1[x], f2[x]}, {x, 0, 5}, Frame → True]
```

Out[67]=



In[68]:=
```
Sol = Solve[f1[x] == f2[x], x]
```

Out[68]= $\left\{\left\{x \to \frac{1}{4}\left(11 - \sqrt{73}\right)\right\}, \left\{x \to \frac{1}{4}\left(11 + \sqrt{73}\right)\right\}\right\}$

```
pp = {x, f1[x]} /. Sol
  (* we get a list of two points from the two solutions *)
```

Out[69]= $\left\{\left\{\frac{1}{4}\left(11 - \sqrt{73}\right), 2 + \frac{3}{4}\left(11 - \sqrt{73}\right)\right\}, \left\{\frac{1}{4}\left(11 + \sqrt{73}\right), 2 + \frac{3}{4}\left(11 + \sqrt{73}\right)\right\}\right\}$

In[70]:=
```
pp // N
```

Out[70]= {{0.613999, 3.842}, {4.886, 16.658}}

In[71]:=
```
Plot[{f1[x], f2[x]}, {x, 0, 5}, Frame → True,
 Epilog → {PointSize[0.03], Point[pp]}, ImageSize → 300]
```

Out[71]=



In[72]:=
```
f1[x_] := x + 4;
f2[x_] := 3 x^3 - 2 x^2 - x + 5;
```

In[75]:=
```
Plot[{f1[x], f2[x]}, {x, -2, 2},
 Frame → True, ImageSize → 300, PlotRange → {0, 10}]
```

Out[75]=



In[76]:=
```
Sol = Solve[f1[x] == f2[x], x] // Simplify
```

Out[76]= $\left\{\{x \to 1\}, \left\{x \to \frac{1}{6}\left(-1-\sqrt{13}\right)\right\}, \left\{x \to \frac{1}{6}\left(-1+\sqrt{13}\right)\right\}\right\}$

In[77]:=
```
f1[x_] := x + 4;
f2[x_] := 2 x^3 - 2 x^2 - x + 5; (* With this function,
Mathematica gets problems... *)
```

In[79]:=
```
Sol = Solve[f1[x] == f2[x], x] // Simplify
```

Out[79]= $\left\{\left\{x \to \frac{1}{6}\left(2+\frac{8\times 2^{2/3}}{\left(-5+3\,i\sqrt{111}\right)^{1/3}}+\left(-10+6\,i\sqrt{111}\right)^{1/3}\right)\right\},\right.$

$\left\{x \to \frac{1}{12}\left(4-\frac{8\times 2^{2/3}\left(1+i\sqrt{3}\right)}{\left(-5+3\,i\sqrt{111}\right)^{1/3}}+i\left(i+\sqrt{3}\right)\left(-10+6\,i\sqrt{111}\right)^{1/3}\right)\right\},$

$\left.\left\{x \to \frac{1}{12}\left(4+\frac{8\,i\,2^{2/3}\left(i+\sqrt{3}\right)}{\left(-5+3\,i\sqrt{111}\right)^{1/3}}+\left(-1-i\sqrt{3}\right)\left(-10+6\,i\sqrt{111}\right)^{1/3}\right)\right\}\right\}$

In[80]:=
```
Sol // FullSimplify
```

Out[80]= {{x → Root[1 - 2 #1 - 2 #1^2 + 2 #1^3 &, 3]},
 {x → Root[1 - 2 #1 - 2 #1^2 + 2 #1^3 &, 1]}, {x → Root[1 - 2 #1 - 2 #1^2 + 2 #1^3 &, 2]}}

In[81]:=
```
% // N
```

Out[81]= {{x → 1.45161}, {x → -0.854638}, {x → 0.403032}}

Exercise 6: Quadratic Equation & more

Part 1

In[106]:=
```
EQ = a x^2 + b x + c == 0
  (* The equation. Make sure all symbols are unused (blue!) *)
```

Out[106]= $c + b\,x + a\,x^2 == 0$

In[107]:=
```
Solutions = Solve[EQ, x]
 (* This gives a list of solutions as replacement rules *)
```

Out[107]= $\left\{\left\{x \to \frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a}\right\}, \left\{x \to \frac{-b + \sqrt{b^2 - 4\,a\,c}}{2\,a}\right\}\right\}$

In[108]:=
```
EQ /. Solutions (* Check that the solution indeed satisfy EQ *)
```

Out[108]= $\left\{c + \frac{b\left(-b - \sqrt{b^2 - 4\,a\,c}\right)}{2\,a} + \frac{\left(-b - \sqrt{b^2 - 4\,a\,c}\right)^2}{4\,a} == 0,\; c + \frac{b\left(-b + \sqrt{b^2 - 4\,a\,c}\right)}{2\,a} + \frac{\left(-b + \sqrt{b^2 - 4\,a\,c}\right)^2}{4\,a} == 0\right\}$

In[109]:=
```
% // Simplify
```

Out[109]= {True, True}

In[110]:=
```
x1 = x /. First[Solutions]
```

Out[110]= $\frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a}$

In[111]:=
```
x2  = x /. Last[Solutions]
```

Out[111]= $\frac{-b + \sqrt{b^2 - 4\,a\,c}}{2\,a}$

In[112]:=
```
EQ /. x → x1 // Simplify
```

Out[112]= True

In[113]:=
```
EQ /. x → x2 // Simplify
```
Out[113]= True

Part 2

In[114]:=
$$EQ3 = a_3\,x^3 + a_2\,x^2 + a_1\,x + a_0 == 0;$$

In[115]:=
```
Solve[EQ3, x] // Simplify
```
Out[115]=
$$\Bigg\{\Bigg\{x \to -\frac{a_2}{3\,a_3} - \left(2^{1/3}\left(-a_2^2 + 3\,a_1\,a_3\right)\right) \Big/ \left(3\,a_3\left(-2\,a_2^3 + 9\,a_1\,a_2\,a_3 - 27\,a_0\,a_3^2 + \sqrt{\left(-4\left(a_2^2 - 3\,a_1\,a_3\right)^3 + \left(2\,a_2^3 - 9\,a_1\,a_2\,a_3 + 27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) + \frac{1}{3\times 2^{1/3}\,a_3}\left(-2\,a_2^3 + 9\,a_1\,a_2\,a_3 - 27\,a_0\,a_3^2 + \sqrt{\left(-4\left(a_2^2 - 3\,a_1\,a_3\right)^3 + \left(2\,a_2^3 - 9\,a_1\,a_2\,a_3 + 27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\Bigg\},$$

$$\Bigg\{x \to -\frac{a_2}{3\,a_3} + \left(\left(1 + i\sqrt{3}\right)\left(-a_2^2 + 3\,a_1\,a_3\right)\right) \Big/ \left(3\times 2^{2/3}\,a_3\left(-2\,a_2^3 + 9\,a_1\,a_2\,a_3 - 27\,a_0\,a_3^2 + \sqrt{\left(-4\left(a_2^2 - 3\,a_1\,a_3\right)^3 + \left(2\,a_2^3 - 9\,a_1\,a_2\,a_3 + 27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) - \frac{1}{6\times 2^{1/3}\,a_3}\left(1 - i\sqrt{3}\right)\left(-2\,a_2^3 + 9\,a_1\,a_2\,a_3 - 27\,a_0\,a_3^2 + \sqrt{\left(-4\left(a_2^2 - 3\,a_1\,a_3\right)^3 + \left(2\,a_2^3 - 9\,a_1\,a_2\,a_3 + 27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\Bigg\},$$

$$\Bigg\{x \to -\frac{a_2}{3\,a_3} + \left(\left(1 - i\sqrt{3}\right)\left(-a_2^2 + 3\,a_1\,a_3\right)\right) \Big/ \left(3\times 2^{2/3}\,a_3\left(-2\,a_2^3 + 9\,a_1\,a_2\,a_3 - 27\,a_0\,a_3^2 + \sqrt{\left(-4\left(a_2^2 - 3\,a_1\,a_3\right)^3 + \left(2\,a_2^3 - 9\,a_1\,a_2\,a_3 + 27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) - \frac{1}{6\times 2^{1/3}\,a_3}\left(1 + i\sqrt{3}\right)\left(-2\,a_2^3 + 9\,a_1\,a_2\,a_3 - 27\,a_0\,a_3^2 + \sqrt{\left(-4\left(a_2^2 - 3\,a_1\,a_3\right)^3 + \left(2\,a_2^3 - 9\,a_1\,a_2\,a_3 + 27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\Bigg\}\Bigg\}$$

In[116]:=
$$EQ4 = a_4\,x^4 + a_3\,x^3 + a_2\,x^2 + a_1\,x + a_0 == 0;$$

In[117]:=
```
Reduce[EQ4, x] // Simplify
```

Out[117]=

$$\left(a_4 == 0 \,\&\& \left(\left(a_3 \neq 0 \,\&\& \left(x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 \,\&, 1] \,||\, x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 \,\&, 2] \,||\, x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 \,\&, 3]\right)\right) \,||\, \left(\left(\left(x == \frac{-a_1 + \sqrt{a_1^2 - 4\, a_0\, a_2}}{2\, a_2} \,||\, a_2 == 0 \,||\, x + \frac{a_1 + \sqrt{a_1^2 - 4\, a_0\, a_2}}{2\, a_2} == 0\right) \,\&\& (a_0 == 0 \,||\, a_1 \neq 0 \,||\, a_2 \neq 0) \,\&\& \left(x + \frac{a_0}{a_1} == 0 \,||\, a_1 == 0 \,||\, a_2 \neq 0\right) \,\&\& \, a_3 == 0\right)\right)\right) \,||$$

$$\left(a_4 \neq 0 \,\&\& \left(x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \,\&, 1] \,|| \, x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \,\&, 2] \,|| \, x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \,\&, 3] \,|| \, x == \text{Root}[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \,\&, 4]\right)\right)$$

Part 3

In[118]:=
```
Table[1 + x^n, {n, 1, 10}]
```

Out[118]=

$$\{1 + x, 1 + x^2, 1 + x^3, 1 + x^4, 1 + x^5, 1 + x^6, 1 + x^7, 1 + x^8, 1 + x^9, 1 + x^{10}\}$$

In[119]:=
```
% // Factor
```

Out[119]=

$$\{1 + x, 1 + x^2, (1 + x)(1 - x + x^2), 1 + x^4, (1 + x)(1 - x + x^2 - x^3 + x^4), (1 + x^2)(1 - x^2 + x^4), (1 + x)(1 - x + x^2 - x^3 + x^4 - x^5 + x^6), 1 + x^8, (1 + x)(1 - x + x^2)(1 - x^3 + x^6), (1 + x^2)(1 - x^2 + x^4 - x^6 + x^8)\}$$

Part 4

In[120]:=
```
RandomReal[{-1, 1}]
```

Out[120]= 0.380791

In[121]:=
```
MyPolynom = Sum[RandomReal[{-1, 1}] x^n, {n, 0, 5}]
```

Out[121]=

$$-0.176801 - 0.328263\, x + 0.109636\, x^2 - 0.273546\, x^3 + 0.314036\, x^4 + 0.15377\, x^5$$

In[122]:=
```
Plot[MyPolynom, {x, -5, 5}, ImageSize → 300, Frame → True, PlotRange → {-5, 5}]
```

Out[122]=



In[123]:=
```
NSolve[MyPolynom == 0, x] (* Numerically, this is no problem *)
```

Out[123]=
```
{{x → -2.83451}, {x → -0.406505}, {x → 0.0325647 - 0.937638 ⅈ},
 {x → 0.0325647 + 0.937638 ⅈ}, {x → 1.13364}}
```

```
NSolve[{MyPolynom == 0 && x ∈ Reals}, x] (* Only real solutions *)
```

Out[124]=
```
{{x → -2.83451}, {x → -0.406505}, {x → 1.13364}}
```

In[125]:=
```
(* For fun: Plot a few more polynoms *)
SEVERAL = Table[Sum[RandomReal[{-1, 1}] x^n, {n, 0, 5}], {i, 1, 4}];
Plot[SEVERAL, {x, -5, 5}, ImageSize → 300,
 Frame → True, PlotRange → {{-3, 3}, {-5, 5}}, PlotStyle → Thick]
```

Out[126]=



Exercise 7: Maximizing the Area of a Rectangle

In[127]:=
```
Clear[a, b, AArea, Periphery, asol, amax, bmax];
(* Make sure these are not used *)
```

In[128]:=
```
Solve[Periphery == 2 a + 2 b, a] // First
 (* get side a for a given b and Periphery *)
```

Out[128]= $\left\{a \to \frac{1}{2}(-2\,b + \text{Periphery})\right\}$

In[129]:=
```
asol = a /. % (* keep the expression for later *)
```

Out[129]= $\frac{1}{2}(-2\,b + \text{Periphery})$

In[130]:=
```
AArea = a b /. a → asol (* express the area as a function of b and Periphery *)
```

Out[130]= $\frac{1}{2}\,b\,(-2\,b + \text{Periphery})$

In[131]:=
```
sol = Maximize[AArea, b] (* find b for maximal area *)
```

Out[131]= $\left\{\frac{\text{Periphery}^2}{16}, \left\{b \to \frac{\text{Periphery}}{4}\right\}\right\}$

In[132]:=
```
bmax = b /. (% // Last) (* remember the solution for b *)
```

Out[132]= $\frac{\text{Periphery}}{4}$

In[133]:=
```
amax = asol /. b → bmax (* get the solution for a *)
```

Out[133]= $\frac{\text{Periphery}}{4}$

In[134]:=
```
amax == bmax (* check that it is a square *)
```

Out[134]= True

In[135]:=
```
TEST = Periphery → 40;
```

In[136]:=
```
Plot[AArea /. TEST, {b, 0, 20}, Frame → True,
 Filling → Axis, GridLines → {{bmax /. TEST}, {}}]
```

Out[136]=



Exercise 8: A real world problem

A silicon wafer of 200mm diameter contains two types of microchips: Chips 'A' with a size of 5x5 mm^2 and Chips 'B' with 3x3 mm^2. The relative amount of chips can be chosen by you. Let us assume that chips 'A' cover an area fraction α of the wafer ($0<=\alpha<=1$).
The vendor produces $N_{wafer} = 12$ wafers . This 'batch' is split in two flavors: A fraction β (i.e. $\beta\ N_{wafer}$ wafers) is produced such that chips 'A' can be used, the remaining wafer are for Chips 'B'.
You need 10000 chips 'A', not more.
How do you chose α and β so that you get as many chips 'B' as possible?

In[82]:=
```
Clear[NumberofWafers, Wafersize, LostFraction, ChipArea];
```

In[84]:=
```
PARAMETERS = {
   NumberOfWafers → 12,
   WaferArea → π (200/2)^2,
   ChipAArea → 3 × 3,
   ChipBArea → 5 × 5,
   NChipA → 10 000
  }; (* define a replacement rule to later exchange variables by values *)
```

In[85]:=
```
$Assumptions =
  WaferArea > 0 && NChipA > 0 && NumberOfWafers > 0 && ChipAArea > 0 && ChipBArea > 0;
```

In[86]:=
```
NA = β NumberOfWafers (α WaferArea)/ChipAArea;

NB = (1 - β) NumberOfWafers ((1 - α) WaferArea)/ChipBArea;
```

In[88]:=
```
{NA, NB} /. PARAMETERS
```

Out[88]= $\left\{\frac{40000\,\pi\,\alpha\,\beta}{3},\ 4800\,\pi\,(1-\alpha)\,(1-\beta)\right\}$

In[89]:=
```
Plot3D[{NA, NB, 10 000} /. PARAMETERS,
 {α, 0, 1}, {β, 0, 1}, PlotRange → {0, 12 000}]
```

Out[89]=




```
βsol = Solve[NA == NChipA, β] // First
  (* This β guarantees NChipA chips for all cases *)
```

Out[90]= $\left\{\beta\to\frac{\text{ChipAArea NChipA}}{\text{NumberOfWafers WaferArea}\ \alpha}\right\}$

In[94]:=
```
Plot[NB /. βsol /. PARAMETERS, (* This is the number of Chips 'B'
  for the β which guarantees the correct number of chips 'A' *)
 {α, 0, 1}, Frame → True, PlotRange → {0, 5000}, Filling → Axis]
```

Out[94]=



In[98]:=
```
D[NB /. βsol, α] // Simplify (* Get the derivative of the above function *)
```

Out[98]=

$$\frac{\text{ChipAArea NChipA} - \text{NumberOfWafers WaferArea}\,\alpha^2}{\text{ChipBArea}\,\alpha^2}$$

In[99]:=
```
Solve[% == 0, α] (* find the α which maximizes it *)
```

Out[99]=

$$\left\{\left\{\alpha \to -\frac{\sqrt{\text{ChipAArea}}\sqrt{\text{NChipA}}}{\sqrt{\text{NumberOfWafers}}\sqrt{\text{WaferArea}}}\right\}, \left\{\alpha \to \frac{\sqrt{\text{ChipAArea}}\sqrt{\text{NChipA}}}{\sqrt{\text{NumberOfWafers}}\sqrt{\text{WaferArea}}}\right\}\right\}$$

In[100]:=
```
αmax = α /. Last[%] (* assign the positive solution *)
```

Out[100]=

$$\frac{\sqrt{\text{ChipAArea}}\sqrt{\text{NChipA}}}{\sqrt{\text{NumberOfWafers}}\sqrt{\text{WaferArea}}}$$

In[102]:=
```
βmax = β /. βsol /. α → αmax
   (* and get the corresponding β. Surprisingly, it's the same as αmax *)
```

Out[102]=

$$\frac{\sqrt{\text{ChipAArea}}\sqrt{\text{NChipA}}}{\sqrt{\text{NumberOfWafers}}\sqrt{\text{WaferArea}}}$$

In[103]:=
```
αmax /. PARAMETERS // N
 (* get the numberical value for our particular parameter set *)
```

Out[103]=
```
0.488603
```

```
{NA, NB} /. α → αmax /. β → βmax // Simplify
 (* and get the number of chips in general *)
```

Out[104]=

$$\left\{\text{NChipA}, \frac{1}{\text{ChipBArea}}\left(\text{ChipAArea NChipA} + \text{NumberOfWafers WaferArea} - 2\sqrt{\text{ChipAArea NChipA NumberOfWafers WaferArea}}\right)\right\}$$

In[105]:=
```
% /. PARAMETERS // N (* and the numerical value, our result *)
```

Out[105]=
```
{10 000., 3943.74}
```

Exercise 9: Area of a Circle

In[137]:=
```
Clear[x, y, r, sol];
```

In[138]:=
```
EQ = x^2 + y^2 == r^2;
```

In[139]:=
```
y[r_, x_] = y /. Solve[EQ, y] // Last
```

Out[139]= $\sqrt{r^2 - x^2}$

In[140]:=
```
Plot[y[1, x], {x, -1.0, 1.0}, PlotRange → {0, 1},
 AspectRatio → 1 / 2, Filling → Axis, Frame → True]
```

Out[140]=


In[141]:=
```
$Assumptions = True;
```

```
Integrate[y[r, x], {x, -r, r}] (* this takes very long on my computer,
but finally gives the correct result *)
```

Out[142]= $\frac{1}{2} \pi r \sqrt{r^2}$

In[143]:=
```
$Assumptions = r > 0; (* Things get simpler with positive r: *)
```

```
Integrate[y[r, x], {x, -r, r}] (* much faster now *)
```

Out[144]= $\frac{\pi r^2}{2}$

Indefinite Integral

```
sol = Integrate[y[r, x], x] (* general solution. This is very quick! *)
```

Out[145]= $\frac{1}{2} \left(x \sqrt{r^2 - x^2} + r^2 \text{ArcTan}\left[\frac{x}{\sqrt{r^2 - x^2}} \right] \right)$

In[147]:=
```
Plot[sol /. r → 1, {x, -1, 1}, Frame → True, Filling → Axis]
```

Out[147]=



In[152]:=
```
sol /. x → r
```

Power: Infinite expression $\frac{1}{\sqrt{0}}$ encountered.

Out[152]= Indeterminate

```
Limit[sol, x → r] (* oups,
this does not work (or at least gives the wrong sign!) *)
```

Out[154]= Indeterminate

In[153]:=
```
Limit[sol, x → r, Direction → 1] (* we must come from the left! *)
```

Out[153]= $\frac{\pi r^2}{4}$

Exercise 10: More Plotting

First Part

In[155]:=
```
SetOptions[Plot, {Frame → True, Filling → Axis}];
```

In[156]:=
```
$Assumptions = {a > 0, λ > 0};
```

In[157]:=
```
f[x_] = Sin[a x] Exp[-λ x];
```

In[158]:=
```
Reduce[f[1] == 0, a] // Simplify (* Find a such that have a zero at x=1 *)
```

Out[158]= $C[1] \in \mathbb{Z}$ && $(a == 2 \pi C[1] \,||\, \pi + 2 \pi C[1] == a)$

In[159]:=
```
% /. C[1] → 0 // Simplify (* pick some constant,
 solution a=0 excluded by $Assumptions *)
```

Out[159]= a == π

In[160]:=
```
Simplify[(f[k] /. a → π) == 0, k ∈ Integers]
(* Check that with a→π, all integer x positions are zeros *)
```

Out[160]= True

In[161]:=
```
Plot[Evaluate[f[x] /. a → π /. λ → {0, 0.5, 1}], {x, 0, 10}, Filling → None]
```

Out[161]=



Saw Tooth

In[162]:=
```
g[x_, k_] = (-1)^k/k Sin[k x];
```

In[163]:=
```
Plot1 = Plot[Table[g[x, k], {k, 1, 4}], {x, -π, 3 π}, ImageSize → 400]
```

Out[163]=



In[165]:=
```
f[x_, k_] := Sum[g[x, j], {j, 1, k}];
```

In[166]:=
```
Plot2 = Plot[f[x, 5], {x, -π, π}, ImageSize → 400]
```

Out[166]=




```
Manipulate[ (* See later for 'Manipulate' *)
 Plot[f[x, k], {x, -π, π}, ImageSize → 400, PlotRange → {-2, 2}]
 , {{k, 1}, 1, 10, 1}]
```

Out[167]=



- **Reverse: Find the Fourier Coefficients for the saw tooth function '-x':**

In[168]:=
```
Integrate[-x/π Sin[k x], {x, 0, π}]
```

Out[168]=

$$\frac{k\,\pi\,\text{Cos}[k\,\pi] - \text{Sin}[k\,\pi]}{k^2\,\pi}$$

In[169]:=
```
Simplify[%, k ∈ Integers]
```
Out[169]=
$$\frac{(-1)^k}{k}$$

Exercise 11: Drawig a Weel

In[170]:=
```
Origin = {0, 0};
```
In[171]:=
```
wheel = Table[Line[{Origin, {Cos[2 π phi], Sin[2 π phi]}}], {phi, 0, 1, 1 / 11}];
```
In[172]:=
```
Show[Graphics[{Thick, Blue, wheel, Circle[Origin, 1]}],
 Frame → True, ImageSize → 300]
```
Out[172]=


Exercise 12: Minimal Interconnect Distance of 4 Points, including Manipulate

The corners are at (0,0), (1,0), ... (1,1)

In[173]:=
```
Clear[x, y, l, p0, p1, p2, p3]
```
We assume that we have a ‘H’ shape which is symmetric. Then there is only one free parameter, the distance of the center point at position x (from the left) and 1-x

In[174]:=
```
l[x_] = (1 - 2 x) + 4 Sqrt[(1/2)^2 + x^2] ;(* central part + 4 time the arms *)
```
$$l[x\_] = (1 - 2x) + 4\sqrt{\left(\frac{1}{2}\right)^2 + x^2}$$

In[175]:=
```
{l[1/2] == 2 √2, l[0] == 3} (* Check tow obvious values *)
```
Out[175]=
```
{True, True}
```
In[176]:=
```
Minimize[l[x], x]
```
Out[176]=
$$\left\{1+\sqrt{3}, \left\{x \to \frac{1}{2\sqrt{3}}\right\}\right\}$$

In[177]:=
```
xmin = x /. Last[%];
```
In[178]:=
```
xmin // N
```
Out[178]=
```
0.288675
```
In[179]:=
```
Plot[l[x], {x, 0, 1 / 2}, PlotRange → {0, 3}, Frame → True]
```
Out[179]=



In[180]:=
```
Show[Plot[Evaluate[l[x]], {x, 0, 1 / 2}, PlotRange → {2.5, 3}, Frame → True],
 Graphics[{PointSize[Large], Point[{x, l[x]}] /. x → xmin}]]
(* Add a point to where the minimum is *)
```
Out[180]=



In[181]:=
```
angle = 2 ArcTan[(1/2)/x /. x → xmin] 360/(2 π) (* Show that all angles are equal *)
```
Out[181]=
```
120
```

Do some nice Plotting

In[182]:=
```
p0 = {0, 0};
p1 = {1, 0};
p2 = {1, 1};
p3 = {0, 1};
corners = Point[{p0, p1, p2, p3}];(* corner points *)
```
In[183]:=
```
xl = {x, 1 / 2}; xr = {1 - x, 1 / 2}; (* left and right point *)
```
In[184]:=
```
mygraphic[x_] = Graphics[{
    PointSize[Large],
    corners,
    Line[{p0, xl, p3}], Line[{xl, xr}], Line[{p1, xr, p2}],
    Red,
    Point[{xl, xr}]}
  ];
```
In[186]:=
```
Show[mygraphic[0.1], ImageSize → 200] (* See if it works *)
```
Out[186]=



In[187]:=

```
Manipulate[
 Show[mygraphic[a], Frame → True,
  ImageSize → 200, PlotRange → {{-0.1, 1.1}, {-0.1, 1.1}}],
 {{a, 1 / 4, "dist"}, 0, 1/2, Appearance → "Labeled"}
]
```

Out[187]=



In[194]:=

```
Manipulate[ (* The same with the previous plot *)
 Show[
  GraphicsRow[
   {
    mygraphic[a]
    , Show[{Plot[l[x], {x, 0, 1 / 2}, PlotRange → {{0, 0.5}, {2.5, 3}}],
      Graphics[{PointSize[Large], Red, Point[{x, l[x]}] /. x → a}]}]
   }
   , ImageSize → 500
  ](* end GraphicsRow *)
 ] (* end Show *)
 , {{a, 1 / 4, "dist"}, 0, 1/2, Appearance → "Labeled"}
]
```

Out[194]=



Exercise 13: Car lifter

In[195]:=

```
SetOptions[Plot, {AspectRatio → 1 / GoldenRatio, Frame → True, Filling → Axis}];
```

In[196]:=

```
Solve[(h/2)^2 + (x/2)^2 == l^2, h] (* One edge,
half x and half h for a triangle with 90 degrees at the center *)
```

Out[196]=

$$\left\{\left\{h \to -\sqrt{4\, l^2 - x^2}\right\}, \left\{h \to \sqrt{4\, l^2 - x^2}\right\}\right\}$$

In[197]:=
```
h[x_] = h /. Last[%]
```
Out[197]= $\sqrt{4\, l^2 - x^2}$

In[198]:=
```
Plot[h[x] /. l → 1, {x, 0, 2}]
(* Plot & Check: For x=0, h = 2l, for x=2l, h=0 *)
```
Out[198]=


In[199]:=
```
-h'[x]
```
Out[199]= $\frac{x}{\sqrt{4\, l^2 - x^2}}$

In[200]:=
```
Plot[-1 / h'[x] /. l → 1, {x, 0, 2}, PlotRange → {0, 100}]
(* h-force * h-distance = constant = x-force * x-dist = const * x
     → hforce = const * dx / dh
*)
```
Out[200]=


Make a Graphics of this...

In[201]:=
```
{P1, P2, P3, P4, Pmax} =
  {{0, 0}, {x/2, h[x]/2}, {-x/2, h[x]/2}, {0, h[x]}, {0, h[0]}} /. l → 1;
(* corner points *)
```

In[202]:=
```
G = Graphics[{
     Thick
     , Line[{P1, P2}], Line[{P2, P4}],
     Line[{P4, P3}], Line[{P1, P3}] (* four outer lines *)
     , Red, Line[{P2, P3}] (* center line *)
     (*,Point[Pmax]*) (* top point *)
   }, ImageSize → 200];
```

In[203]:=
```
Manipulate[
 Show[G /. x → a, PlotRange → {{-1.5, 1}, {0, 2}}],
 {{a, 1, "x"}, 0, 2, Appearance → "Labeled"}
]
```

Out[203]=



Exercise 14: Equal Lengths

In[204]:=
```
Clear[x, y, x1, y1, p1, px, pp, ϰ]; (* for safety *)
```

In[205]:=
```
$Assumptions = x > 0 && y > 0 && x1 > 0 && y1 > 0 && ϰ > 0;
```

In[206]:=
```
p1 = {x1, y1}; (* this point is given (the other is the origin) *)
px = {x, y}; (* unknown point: the kink is here *)
```

In[208]:=
```
L0 = Norm[px]; // Simplify (* distance from origin to unknown (kink) point *)
L1 = Norm[p1 - px]; // Simplify
(* distance from unkown point to given point *)
```

In[210]:=
```
Solve[L0 == L1 && 2 L0 == κ, {x, y}] // Last //
 FullSimplify (* equal lengths && sum == κ *)
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[210]=
$$\left\{x \to \frac{1}{2}\left(x1 - y1\sqrt{-1 + \frac{\kappa^2}{x1^2 + y1^2}}\right),\ y \to \frac{1}{2}\left(y1 + x1\sqrt{-1 + \frac{\kappa^2}{x1^2 + y1^2}}\right)\right\}$$

In[211]:=
```
pp[x1_, y1_, κ_] = {x, y} /. % (* define a fucntion *)
```

Out[211]=
$$\left\{\frac{1}{2}\left(x1 - y1\sqrt{-1 + \frac{\kappa^2}{x1^2 + y1^2}}\right),\ \frac{1}{2}\left(y1 + x1\sqrt{-1 + \frac{\kappa^2}{x1^2 + y1^2}}\right)\right\}$$

In[212]:=
```
Manipulate[
 Graphics[Line[{{0, 0}, pp[First[pt], Last[pt], 3], pt}],
  PlotRange → 2, ImageSize → 200]
 , {{pt, {0, 1}}, Locator}
]
```

Out[212]=



Exercise 15: Adventskalender

In[213]:=
```
Clear[dϕ, Rout]
```

In[214]:=
```
dϕ[n_] = 2π/n;(* Angle between two corners *)
```

In[215]:=
```
Rout[n_] = 1/Cos[dϕ[n] / 2];(* For plotting: radius to OUTER points *)
```

In[216]:=
```
Corner[k_, n_] = Rout[n] * {Cos[k dϕ[n]], Sin[k dϕ[n]]};
(* For plotting: coordinates of k-th corner *)
```

In[217]:=
```
{Corner[1, 10], Corner[1 + 1, 10]} // N (* Check that expression is ok *)
```

Out[217]=
```
{{0.850651, 0.618034}, {0.32492, 1.}}
```

In[218]:=
```
Show[
 Graphics[{
   Blue, Circle[{0, 0}, 1]
   , Black, Thick, Table[Line[{Corner[k, 5], Corner[k + 1, 5]}], {k, 1, 5}]
  }]
 , ImageSize → 200
 , Frame → True
 , PlotRange → {{-1.2, 1.2}, {-1.2, 1.2}}
]
```

Out[218]=



- **Now solve the exercise**

In[220]:=
```
Clear[L]
```

In[221]:=
```
L[n_] = n 2 Tan[dϕ[n]/2];(* This is the result we are looking for !*)
```

In[222]:=
```
Limit[L[n], n → ∞]
```
Out[222]= 2π

In[223]:=
```
Plot[{L[n]/(2 π), 1.001, 1}, {n, 2, 200}, Frame → True]
(* Compare to 2π. Also plot 1, and 1 %% *)
```
Out[223]=



In[224]:=
```
NSolve[L[n] == 1.001 × 2 π, n, Reals] (* This does not find a solution !*)
```
Out[224]= NSolve$\left[2\, n \tan\left[\frac{\pi}{n}\right] == 6.28947, n, \mathbb{R}\right]$

In[225]:=
```
FindRoot[L[n] - 1.001 × 2 π, {n, 10}] (* Starting at N=1 does not work !*)
```
Out[225]= {n → 57.3918}

Exercise 16: Particle Absorption

- Particles are absorbed in a scintillator with function f[x]=A Exp[-α x]
- We want to compose a scintillator of 2 layers of thickness T1, T2 such that the number of absorbed particles is the same in both layers.
- Total thickness is T = T1+T2

In[226]:=
```
$Assumptions = α > 0;
```

In[227]:=
```
Integrate[A Exp[-α x], {x, 0, ∞}]
```
Out[227]= $\frac{A}{\alpha}$

In[228]:=
```
Solve[% == 1, A] // First
```
Out[228]= $\{A \to \alpha\}$

```
fabs[x_] = A Exp[-α x] /. % (* fabs is now normalize *)
```
Out[229]= $e^{-x\alpha}\alpha$

In[230]:=
```
Integrate[fabs[x], {x, 0, ∞}]
```
Out[230]= 1

In[231]:=
```
Fabs[t1_, t2_] = Integrate[fabs[x], {x, t1, t2}]
  (* Absorptions betwwen t1 and t2 *)
```
Out[231]= $e^{-t1\,\alpha} - e^{-t2\,\alpha}$

In[233]:=
```
Limit[Fabs[0, t], t → ∞] (* Just a check *)
```
Out[233]= 1

```
EQ = Fabs[0, T1] == Fabs[T1, T]
   (* Now solve the problem: Set same absorptions in both layers *)
```
Out[234]= $1 - e^{-T1\,\alpha} == -e^{-T\,\alpha} + e^{-T1\,\alpha}$

In[235]:=
```
Solve[EQ, T1]
```
Out[235]= $\left\{\left\{T1 \to \text{ConditionalExpression}\left[\frac{2\,i\,\pi\,C[1] + \text{Log}\left[\frac{2\,e^{T\alpha}}{1+e^{T\alpha}}\right]}{\alpha},\ C[1] \in \mathbb{Z}\right]\right\}\right\}$

In[236]:=
```
Sol = % /. C[1] → 0 // First
   (* We find more solutions than we need. Use the first *)
```
Out[236]= $\left\{T1 \to \frac{\text{Log}\left[\frac{2\,e^{T\alpha}}{1+e^{T\alpha}}\right]}{\alpha}\right\}$

In[237]:=
```
T1sol = T1 /. Sol /. T → 1 (* This is the solution T1 for T=1 *)
```
Out[237]= $\frac{\text{Log}\left[\frac{2\,e^{\alpha}}{1+e^{\alpha}}\right]}{\alpha}$

In[244]:=
```
T1sol /. α → 0.0000001
```
Out[244]= 0.5

In[245]:=
```
Limit[T1sol, α → 0] // FullSimplify
 (* Tough. Mathematica does not find the exact solution *)
```
Out[245]= $\lim_{\alpha \to 0} \frac{\alpha + \text{Log}[2] - \text{Log}[1 + e^{\alpha}]}{\alpha}$

In[246]:=
```
Plot[T1sol, {α, 0, 10}, Frame → True, Filling → Axis, PlotRange → {0, 0.5}]
```
Out[246]=